# Maintainer: Sébastien Luttringer <seblu@archlinux.org>
# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: Andreas Radke <andyrtr@archlinux.org>

pkgname=which
pkgver=2.21
pkgrel=6
pkgdesc='A utility to show the full path of commands'
arch=('x86_64')
url='https://savannah.gnu.org/projects/which/'
license=('GPL3')
depends=('glibc' 'bash')
# gpg key is using deprecated md5 algo, do not use
# check if a new one is issued in the next release
source=("https://ftp.gnu.org/gnu/$pkgname/$pkgname-$pkgver.tar.gz")
sha256sums=('f4a245b94124b377d8b49646bf421f9155d36aa7614b6ebf83705d3ffc76eaad')

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
